Equilibrium of an Electrodynamic Cable Propelling an Atmosphere Harvestor

G. Flanagan
Alna Space Program

September 27, 2011: Revision October 11 2011

Introduction

A current passing through an orbital cable will interact with the earth's magnetic field through the electromotive relation $\overline{F}$=J × $\overline{B}$, where J is the current and $\overline{B}$ is the magnetic field vector. One application of this propulsive force is to drag a mass collector through the upper atmosphere. The cable constantly makes up the momentum loss that comes from collecting gas. A possible system would involve a collector at the bottom of a long cable, and a solar power station at the upper end of the cable. The current loop would be completed with electron collection at one end, and an electron gun at the other. An example of a similar application for electrodynamic tether propulsion is to reboost the International Space Station [1]. Some additional system details of a tether propulsion system are described in a related patent [2]. The present study gives a solution approach to determine the steady-state conditions and cable trajectory for a system that drags a collector through the atmosphere. Functions are provided that allow one to determine many of the cable system parameters needed to meet a mass collection goal.

In order to simplify the solution, we will limit the application to circular orbits at the equatorial plane.

Revision Notes

This notebook was originally released without cable aerodynamic drag being implemented in the solution functions. This version uses a new atmospheric model (spaceLaunch`atmosphereModel) that works for greater altitudes than the previously used U.S. Standard Model, and drag has been incorporated into the solution function.

Solution Approach

Derivation of Governing Differential Equations

[Author's note: I have decided to show the *Mathematica* steps involved with manipulating the differential equations, instead of cutting and pasting dead equations in a more traditional development. The derivation will "play" in *Mathematica*, allowing the reader to change initial assumptions and force models. I don't know if this enlightens or obfuscates, but it is in keeping with the Alna Space Program philosophy of sharing everything].

Figure 1 shows the coordinate system that will be used, and defines some geometric relations. s is the path length along the cable. R and ψ define the location of a point relative to the earth's center. α is the local angle the cable makes relatives to the angular coordinate (normal to $\vec{R}$). The points A and B are the ends of the cable which terminate into masses.



Figure 1. Definitions and Coordinates for Cable



Figure 2 Differential Element of Cable

We take the sum of all the forces in the θ direction. The magnetic force is normal to the current and therefore normal to the cable. We will use a simple model for the earth's magnetic field with the field dropping off with the cube of the distance from the surface, or B[R]= $\frac{J\,B_e\,R_e^3}{R(s)^3}$ where J is the current and B_e is the field strength at the surface. This assumes that the cable is orbiting at the equator so there is no change in the field strength due to orbital inclination. The aerodynamic drag on the cable uses the model $F_{\text{aero}} = \frac{1}{2}\, C_D\, \rho_{\text{aero}}\, A\, V^2$, where C_D is the coefficient of drag. The projected area A of the cable segment is d Sin[α] ds where d is the cable diameter. The local velocity is simply ω R, where ω is the orbital angular velocity. T is the cable tension. Putting these forces together with appropriate angle transformations gives

```
-T[s] Cos[α[s]] + (T[s] + T'[s] ds) Cos[α[s] + α'[s] ds] + J Bₑ Rₑ^3 / R[s]^3 Sin[α[s]] ds -
 1 / 2 C_D d Sin[α[s]] ρ_air[R[s]] (ω R[s])^2 ds
```

$$\frac{ds\, J\, Sin[\alpha[s]]\, B_e\, R_e^3}{R[s]^3} - Cos[\alpha[s]]\, T[s] - \frac{1}{2}\, d\, ds\, \omega^2\, R[s]^2\, Sin[\alpha[s]]\, C_D\, \rho_{air}[R[s]] + Cos[\alpha[s] + ds\, \alpha'[s]]\, (T[s] + ds\, T'[s])$$

Expanding the compound trigonometric terms and applying small angle assumptions (note that in *Mathematica* the symbol % means to use the previous result)

```
TrigExpand[%] /. {Cos[ds α′[s]] → 1, Sin[ds α′[s]] → ds α′[s]}
```

$$\frac{ds\, J\, Sin[\alpha[s]]\, B_e\, R_e^3}{R[s]^3} - \frac{1}{2}\, d\, ds\, \omega^2\, R[s]^2\, Sin[\alpha[s]]\, C_D\, \rho_{air}[R[s]] + ds\, Cos[\alpha[s]]\, T'[s] - ds\, Sin[\alpha[s]]\, T[s]\, \alpha'[s] - ds^2\, Sin[\alpha[s]]\, T'[s]\, \alpha'[s]$$

Finally, take ds^2 is small compared to ds, and dividing through by ds.

```
sumTheta = Simplify[(% /. {ds^2 → 0}) / ds]
```

$$\frac{J\, Sin[\alpha[s]]\, B_e\, R_e^3}{R[s]^3} - \frac{1}{2}\, d\, \omega^2\, R[s]^2\, Sin[\alpha[s]]\, C_D\, \rho_{air}[R[s]] + Cos[\alpha[s]]\, T'[s] - Sin[\alpha[s]]\, T[s]\, \alpha'[s]$$

Next, consider the sum of forces in the R direction. In this case, the acceleration due to gravity and the acceleration due to the orbital velocity are included. $\overline{\rho}$ is the linear density of the cable (kg/m), and μ is the gravitational constant times the mass of the earth.

```
-T[s] Sin[α[s]] + (T[s] + T'[s] ds) Sin[α[s] + α'[s] ds] - ρ̄ (μ / R[s]^2 - ω^2 R[s]) ds -
 J Bₑ Rₑ^3 / R[s]^3 Cos[α[s]] ds
```

$$-ds\, \overline{\rho} \left(\frac{\mu}{R[s]^2} - \omega^2\, R[s]\right) - \frac{ds\, J\, Cos[\alpha[s]]\, B_e\, R_e^3}{R[s]^3} - Sin[\alpha[s]]\, T[s] + Sin[\alpha[s] + ds\, \alpha'[s]]\, (T[s] + ds\, T'[s])$$

Go through the same transformations as used for the θ equations.

```
TrigExpand[%] /. {Cos[ds α′[s]] → 1, Sin[ds α′[s]] → ds α′[s]}
```

$$-\frac{ds\, \mu\, \overline{\rho}}{R[s]^2} + ds\, \omega^2\, \overline{\rho}\, R[s] - \frac{ds\, J\, Cos[\alpha[s]]\, B_e\, R_e^3}{R[s]^3} + ds\, Sin[\alpha[s]]\, T'[s] + ds\, Cos[\alpha[s]]\, T[s]\, \alpha'[s] + ds^2\, Cos[\alpha[s]]\, T'[s]\, \alpha'[s]$$

```
sumR = Simplify[(% /. {ds^2 → 0}) / ds]
```

$$\overline{\rho} \left(-\frac{\mu}{R[s]^2} + \omega^2\, R[s]\right) - \frac{J\, Cos[\alpha[s]]\, B_e\, R_e^3}{R[s]^3} + Sin[\alpha[s]]\, T'[s] + Cos[\alpha[s]]\, T[s]\, \alpha'[s]$$

For steady-state operation, the sum of these forces should equal zero. It is possible to manipulate these equilibrium equations to obtain and equations that separate the tension and angle derivatives. These are the forms that will be used in the numerical integration.

```
Solve[Eliminate[{sumR == 0, sumTheta == 0}, α'[s]], T'[s]][[1, 1]] // Simplify
```

$$T'[s] \to \frac{1}{2\, R[s]^2}\, Sin[\alpha[s]] \left(2\, \overline{\rho} \left(\mu - \omega^2\, R[s]^3\right) + d\, \omega^2\, Cos[\alpha[s]]\, R[s]^4\, C_D\, \rho_{air}[R[s]]\right)$$

```
Solve[Eliminate[{sumR == 0, sumTheta == 0}, T'[s]], α'[s]][[1, 1]] // Simplify
```

$$\alpha'[s] \to \left(4\,\mathrm{Cos}[\alpha[s]]\,\bar{\rho}\,R[s]\,\left(\mu - \omega^2 R[s]^3\right) + 4\,J\,B_e\,R_e^3 - 2\,d\,\omega^2 R[s]^5\,\mathrm{Sin}[\alpha[s]]^2\,C_D\,\rho_{air}[R[s]]\right) \Big/ \left(4\,R[s]^3\,T[s]\right)$$

Boundary Conditions

The governing system is a pair of first order differential equations requiring boundary conditions for T and α. However, the cable has two ends and we would like to specify T and α at both ends. The approach must be to specify boundary conditions at one end and then determine the values at the far end that are consistent with the input conditions. It is convenient to specify conditions at the lower end of the cable; point A. At point A one knows the mass of the atmosphere collection device and therefore the cable tension and angle, as shown in Figure 3, where F_A is the drag force from the collection device, and M_A is the device mass. From the figure,

$$T = \sqrt{F_A^2 + \left(M_A\left(\frac{\mu}{R^2} - R\,\omega^2\right)\right)^2}$$

and

$$\alpha = \tan^{-1}\left(\frac{M_A\left(\frac{\mu}{R^2} - R\,\omega^2\right)}{F_A}\right)$$



Figure 3. Boundary Conditions at Point A

At point B, the desired conditions are

$$T = M_B\left(\frac{\mu}{{R_B}^2} - R_B\,\omega^2\right)$$

$$\alpha = 0$$

The condition α=0 comes from the observation that the top mass cannot be accelerating in the θ direction.

Algorithm

The parameters ω and J must be determined to complete the solution. The current, J, can be determined by integrating the net θ direction forces. For the condition that there is no θ acceleration, we have

$$\int_{R_A}^{R_B} \frac{J\,R_e^3}{r^3}\,dr = \int_{R_A}^{R_B} \frac{1}{2}\,d\,r^2\,\omega^2\,C_D\,\rho_{air}(r)\,dr + F_A$$

Giving

$$J = \frac{B_e \left(\int_{R_A}^{R_B} \frac{1}{2}\, d\, r^3\, \omega^2\, C_D\, \rho_{\text{air}}[r]\, dr + F_A\right) \left(-R_A^2 + R_B^2\right) R_e^3}{2\, R_A^2\, R_B^2}$$

ω is the orbital angular velocity for the center-of-mass of the system. Formally, the center-of-mass is at

$$R_{\text{cm}} = \frac{\int_0^L \overline{\rho}\, R(s) \cos(\alpha(s))\, ds + M_A\, R_A + M_B\, R_B}{L\, \overline{\rho} + M_A + M_B}$$

where L is the length of the cable. R[s] can be computed from α[s] using

$$R'(s) = \sin(\alpha(x))$$

with the boundary condition $R[0] = R_A$

For plotting purposes, we'll also need to determine ψ[s]. From geometry,

$$\psi'(s) = \frac{\cos(\alpha(x))}{R(x)}$$

with the boundary condition ψ[0]=0.

Then

$$\omega = \sqrt{\frac{\mu}{R_{\text{cm}}{}^3}}$$

The system is still not solvable in closed form because either R_B or L must still be determined, that is R_B=R[L]. A good approximation for the solutions of interest is to assume that the cable is close to vertical so that $R[s] \approx R_A + s$, and consequently, $R_B \approx R_A + L$. Using this approximation, and assuming that the cable linear density is constant with length,

$$R_{\text{cm}} = \frac{M_A\, R_A + M_B\, R_B}{M_A + M_B}$$

The final missing piece is to determine a compatible M_B, the mass at point B. From the boundary condition discussion,

$$M_B = T[L] \Big/ \left(\frac{\mu}{R_B{}^2} - R_B\, \omega^2\right)$$

The suggested algorithm is to iteratively solve for M_B. It has been found that an initial guess of $M_B = M_A$ is a good starting point. One goes through the solution to obtain T[L] and ω which are then used to update M_B.

In the boundary condition discussion it is states that α[L]=0. If we correctly determined current such that the overall equilibrium is satisfied, then this condition should be automatically satisfied. A final check of the angle at point B can be used to measure solution error.

Functions

Load the atmosphere model.

```
<<spaceLaunch`atmosphereModel`
```

Functions have been programmed to implement the solution described in the sections above. For the present, we ignore drag on the cable ($C_D = 0$). The drag can be added easily, but I do not currently have a *Mathematica* function for air density at altitudes greater than 80 km. Also, we will use the approximation that $R(s) = R_A + s$. One could iterate further to get a more accurate determination of the center-of-mass location, but for the problems of interest this does not seem to be necessary.

In typical use, there is a single top - level solution function, cableSolution. The input arguments are cableSolution[Ma, Mb, h, L, Fdrag, ρ,dia,coeffDrag] where Ma - mass at the lower termination point (kg), Mb - initial guess for mass at the upper termination point (kg), h - altitude of the lower termination, L - cable length (m), Fdrag - aerodynamic drag force on the mass collector at the lower point (N), ρ - linear density of the cable (kg/m). dia is the cable diameter, and coeffDrag is the coefficient of drag for the cable. There is an option exoatmosphereTemperature. The option allows the user to set the exoatmosphere temperture which will affect the air density at extreme altitudes. The possible values allowed by the atmosphere model are 500, 1000, 1500 and 2000 K. The default is 1000.

The function returns a nested list. The first list element echoes the input and adds some additional scalar values. The list contains {Ma, Mrefined, Rbottom, L, Fdrag, ρ, J, ω}, where Mrefined is the updated compatible mass at point B, Rbottom is the lower radius (h+rearth), J is the current required for steady-state operation, and ω is the orbital angular velocity. The second list element contains the interpolation functions for {T, α, R, ψ}, where T is the cable tension (N), α is the cable angle relative to the θ coordinate, R is the radius, and ψ is the angular coordinate of the cable relative to the earth center. Each of these functions are in terms of the length parameter, s.

```
Protect[exoatmosphereTemperature];Options[cableSolution]={exoatmosphereTemperature→10(
```

```
cableSolution[Ma_,Mb_,h_,L_,Fdrag_,ρ_,dia_,coeffDrag_,opts___]:=Module[{Mrefined,J,ω,
(* Iterate to determine a compatible mass at point B *)
Mrefined=compatibleTopMass[Ma,h,L,Fdrag,ρ,dia,coeffDrag,opts][Mb];
Rbottom=rearth+h;
(* Estimate of the cm location assuming a straight cable *)
cm=(Rbottom*Ma+(Rbottom+L)*Mb)/(Ma+Mb);
ω=Sqrt[μ/cm^3];
(* Compute the current required *)
temp=exoatmosphereTemperature/.{opts}/.Options[cableSolution];
cableDrag=NIntegrate[atmosphereDensityJa77["Total", temp][(r - rearth)/1000]*ω^2*r^2*c
J = (Fdrag + cableDrag)/((BRearth/2)*(1/Rbottom^2 - 1/(L + Rbottom)^2));
{α,R,T}=shapeSolve[Ma,Mrefined,Rbottom,L,Fdrag,ρ,dia,coeffDrag,J,ω,opts];
(* Solve for the cable angular position *)
p=ψ/.First@NDSolve[{ψ'[s]==Cos[α[s]]/R[s],ψ[0]==0},ψ,{s,0,L}];
{{Ma,Mrefined,Rbottom,L,Fdrag,ρ,J,ω},{T,α,R,p}}]
```

Some convenient plotting functions. Each of these will take the result from cableSolution as an input.

```
plotCableTension[ans_]:=Plot[ans[[2,1]][s*1000],{s,0,ans[[1,4]]/1000},
PlotRange→All,AxesLabel→{"s, km","Tension, N"}];

plotCableAngle[ans_]:=Plot[ans[[2,2]][s*1000]/Degree,{s,0,ans[[1,4]]/1000},PlotRange→A
,AxesLabel→{"s, km","α, deg"}];

plotCablePath[ans_]:=Show[{Graphics[Circle[{0,0},rearth/1000,{7Pi/16,9Pi/16}]],
ParametricPlot[{ans[[2,3]][s] Sin[ans[[2,4]][s]]/1000,ans[[2,3]][s] Cos[ans[[2,4]][s]]
{s,0,ans[[1,4]]},Axes→False]}];
```

Utility for checking validity of end conditions. Returns the nested list {{T[0], α[0], h[0], ψ[0]},{T[L], α[L], h[L], ψ[L]}}, where h[s]=R[s]-Rearth. Angles are shown in degrees

```
endConditions[ans_]:=Module[{x},
x=ans[[1,4]];
{{ans[[2,1]][0],ans[[2,2]][0]/Degree,ans[[2,3]][0]-rearth,ans[[2,4]][0]/Degree},
{ans[[2,1]][x],ans[[2,2]][x]/Degree,ans[[2,3]][x]-rearth,ans[[2,4]][x]/Degree}}]
```

A couple of convenient formatting functions

```
tabulateEndConditions[ans_]:=TableForm[Chop@endConditions[ans],TableHeadings→{{"Lower"
```

{Ma, Mrefined, Rbottom, L, Fdrag, ρ, J, ω}

```
tabulateCableData[ans_]:=TableForm[{ans[[1,1]],ans[[1,2]],(ans[[1,3]]-rearth)/1000,ans
ans[[1,5]],ans[[1,6]],ans[[1,7]],ans[[1,8]]},
TableHeadings→{{"Bottom Mass (kg)","Top Mass (kg)","Collector Alt (km)","Cable Length
"Collector Drag (N)","Cable Density (kg/m)","Current (A)","Orbit Ang Vel (Rad/s)"}}]
```

The following are lower level functions that the user would not normally need to call. shapeSolve performs a numerical solution of the governing equilibrium equations for T and α, and in addition solves for R[s]. First six parameters are the same as for cableSolution. J is the cable current, and ω is the orbital angular velocity.

```
shapeSolve[Ma_,Mb_,Rbottom_,L_,Fdrag_,ρ_,dia_,coeffDrag_,J_,ω_,opts___]:=Module[{α,R,
{α,R,T}/.First@NDSolve[{
temp=exoatmosphereTemperature/.{opts}/.Options[cableSolution];
 T'[s]==(1/(2*R[s]^2))*(Sin[α[s]]*(2*ρ*(μ - ω^2*R[s]^3) +
     coeffDrag*dia*ω^2*Cos[α[s]]*R[s]^4*atmosphereDensityJa77["Total", temp][
       (-rearth + R[s])/1000])),
 α'[s]==(1/(4*R[s]^3*T[s]))*
   (4*BRearth*J + 4*ρ*Cos[α[s]]*R[s]*(μ - ω^2*R[s]^3) -
    2*coeffDrag*dia*ω^2*R[s]^5*Sin[α[s]]^2*atmosphereDensityJa77["Total", temp][
      (-rearth + R[s])/1000]),
 R'[s]==Sin[α[s]],
 α[0]== ArcTan[Ma(μ/Rbottom^2-ω^2Rbottom)/Fdrag],
 T[0]==Sqrt[Ma^2(μ/Rbottom^2-ω^2 Rbottom)^2+Fdrag^2],
 R[0]==Rbottom
 },{T,α,R},{s,0,L}]]
```

topmass takes an estimate of the upper mass, Mb, and returns an improved estimate.

```
topMass[Ma_,h_,L_,Fdrag_,ρ_,dia_,coeffDrag_,opts___][Mb_?NumberQ]:=
Module[{J,cm,ω,Rbottom,tmp,a,rr,tt,cableDrag,temp},
temp=exoatmosphereTemperature/.{opts}/.Options[cableSolution];
Rbottom=rearth+h;
cm=(Rbottom*Ma+(Rbottom+L)*Mb)/(Ma+Mb);
ω=Sqrt[μ/cm^3];
(* Next two expressions assume a straight, vertical cable *)
cableDrag=NIntegrate[atmosphereDensityJa77["Total", temp][(r - rearth)/1000]*ω^2*r^2*c
J=(Fdrag + cableDrag)/((BRearth/2)*(1/Rbottom^2 - 1/(L + Rbottom)^2));
{a,rr,tt}=shapeSolve[Ma,Mb,Rbottom,L,Fdrag,ρ,dia,coeffDrag,J,ω,opts];
-tt[L]/((μ/rr[L]^3-ω^2)rr[L])]
```

```
cableTotalDrag[h_,L_,dia_,coeffDrag_,ω_,opts___]:=Module[{temp,Rbottom},
temp=exoatmosphereTemperature/.{opts}/.Options[cableSolution];Rbottom=rearth+h;
NIntegrate[atmosphereDensityJa77["Total", temp][(r - rearth)/1000]*ω^2*r^2*dia*coeffDr
```

compatibleTopMass perform the numerical search for an upper mass that is consistent with the cable tension. Mstart is the initial guess for the mass at point B.

```
compatibleTopMass[Ma_,h_,L_,Fdrag_,ρ_,dia_,coeffDrag_,opts___][Mstart_]:=
Module[{Mb},Mb/.FindRoot[topMass[Ma,h,L,Fdrag,ρ,dia,coeffDrag,opts][Mb]-Mb,{Mb,Mstart
```

The symbols μ, rearth, and BRearth are global constants. BRearth is the product of $B_E\ R_E{}^3$. Assume that the magnetic field at the surface of the earth along the equator is 35 μTesla, taken from Ref [3]. These constants determine the unit system for the function input.

```
Needs["PhysicalConstants`"];
unitStrip[x_]:=Select[x,NumberQ];
Unprotect[BRearth,μ,rearth];
μ=unitStrip[GravitationalConstant*EarthMass];
rearth=EarthRadius//unitStrip;
BRearth=35.*10^-6*rearth^3;
Protect[BRearth,μ,rearth];
```

Testing

Case 1

Consider a cable/collector system with the following input parameters.

```
cableLength = 500 × 10^3;
topMassGuess = 500;
bottomMass = 500;
collectorAltitude = 200 × 10^3;
collectorDrag = 1000;
cableDensity = 0.1;
cableDiameter = .01;
coeffDrag = 1.;
```

Solve the system of equations

```
ans = cableSolution[bottomMass, topMassGuess, collectorAltitude, cableLength,
  collectorDrag, cableDensity, cableDiameter, coeffDrag]
```

```
{{500, 462.08, 6 578 140, 500 000, 1000, 0.1, 70.1201, 0.00111915},
 {InterpolatingFunction[{{0., 500 000.}}, <>],
  InterpolatingFunction[{{0., 500 000.}}, <>], InterpolatingFunction[
   {{0., 500 000.}}, <>], InterpolatingFunction[{{0., 500 000.}}, <>]}}
```

Echo the inputs and calculated values.

```
tabulateCableData[ans]
```

Bottom Mass (kg)	500
Top Mass (kg)	462.08
Collector Alt (km)	200
Cable Length (km)	500
Collector Drag (N)	1000
Cable Density (kg/m)	0.1
Current (A)	70.1201
Orbit Ang Vel (Rad/s)	0.00111915

Tabulate the end conditions. Note that α at the upper point is close to the desired 90 degree value.

tabulateEndConditions[**ans**]

	Tension	α	h	ψ
Lower	1112.62	26.0014	200 000.	0
Upper	2233.23	89.719	693 909.	0.343095

For this example, the cable path is close to vertical.

plotCablePath[**ans**]



Plot the cable tension and path angle.

plotCableTension[**ans**]



plotCableAngle[**ans**]



Case 2

Repeat the previous case, but with a substantially higher drag force. The resulting system is unrealistic, but the example pushes the solution to show a cable path that is significantly non-vertical.

```
cableLength = 500 × 10^3;
topMassGuess = 1000;
bottomMass = 1000;
collectorAltitude = 200 × 10^3;
collectorDrag = 5000;
cableDensity = 0.1;
cableDiameter = .01;
coeffDrag = .1;
```

```
ans = cableSolution[bottomMass, topMassGuess, collectorAltitude, cableLength,
  collectorDrag, cableDensity, cableDiameter, coeffDrag]
```

```
{{1000, 655.138, 6 578 140, 500 000, 5000, 0.1, 349.636, 0.00111915},
 {InterpolatingFunction[{{0., 500 000.}}, <>],
  InterpolatingFunction[{{0., 500 000.}}, <>], InterpolatingFunction[
   {{0., 500 000.}}, <>], InterpolatingFunction[{{0., 500 000.}}, <>]}}
```

The current required to generate this much force is not realistic when compared to the cable density. This is part of the system design iteration that would need to be performed to develop a practical system.

```
tabulateCableData[ans]
```

Bottom Mass (kg)	1000
Top Mass (kg)	655.138
Collector Alt (km)	200
Cable Length (km)	500
Collector Drag (N)	5000
Cable Density (kg/m)	0.1
Current (A)	349.636
Orbit Ang Vel (Rad/s)	0.00111915

This time, α at the upper point is starting to deviate significantly from 90°, indicating that the vertical cable assumptions build into calculating the current and center-of-mass are less accurate.

```
tabulateEndConditions[ans]
```

	Tension	α	h	ψ
Lower	5094.28	11.04	200 000.	0
Upper	8585.38	87.9242	665 522.	1.06828

From the cable path plot, one can see that there is more deviation from the vertical, as would be expected to accommodate the higher drag force.

```
plotCablePath[ans]
```



References

1. Connie Carrington, Vernon Keller, “Orbiter-Towed Reboost for ISS”, presented at Tether Technology Interchange Meeting, Huntsville, Alabama

September 9-10, 1997, NASA / CP-1998- 206900.

2. Johnson et al, United States Patent US 7,118,074 B1, Oct 10, 2006.

3. Tethers in Space Handbook Second Edition, Volume 4, Section 4.6.2, “Electrodynamic Orbit Changes”.